<%
aspinfo()

Sub aspinfo()
    Dim strVariable, strASPVersion
    Dim strCookie, strKey, strSession
    'Retrieve the version of ASP
    strASPVersion = ScriptEngine & " Version " & _
    ScriptEngineMajorVersion & "." & _
    ScriptEngineMinorVersion
%>
	<div id="position">
      &raquo; <a href="admin.asp" class="testo">Control Center</a> &raquo;<%=traducisys("ling_occ_262")%>
</div>
<div id="page" style="text-align:center">
    <table width="100%" border="0" bgcolor="#000000" cellspacing="1" cellpadding="3">
    <tr>
    	<td align="center" valign="top" bgcolor="#99CDFF" colspan="2">
    		<h3>ASP (<%= strASPVersion %>)</h3>
    	</td>
    </tr>
    </table>
    <br />
    <hr />
    <br />
    <h3>Server Variables</h3>
    <table width="100%" border="0" bgcolor="#000000" cellspacing="1" cellpadding="3">
    <%
    For Each strVariable In Request.ServerVariables
			Response.write("<tr>")
			Response.write("<th width=""30%"" bgcolor=""#99CDFF"" align=""left"">" & strVariable & "</th>")
			Response.write("<td bgcolor=""#CBF3FF"" align=""left"">" & Replace(Request.ServerVariables(strVariable),"&", "&amp;") & " </td>")
			Response.write("</tr>")
    Next 'strVariable
    %>
			</table>
			<br />
			<hr />
			<br />
			<h3>Cookies</h3>
			<table width="100%" border="0" bgcolor="#000000" cellspacing="1" cellpadding="3">
    <%
    For Each strCookie In Request.Cookies
    	if Not Request.Cookies(strCookie).HasKeys then
    		Response.write("<tr>")
    		Response.write("<th width=""30%"" bgcolor=""#99CDFF"" align=""left"">" & strCookie & "</th>")
    		Response.write("<td bgcolor=""#CBF3FF"" align=""left"">" & Request.Cookies(strCookie) & " </td>")
    		Response.write("</tr>")
    	Else
    		For Each strKey In Request.Cookies(strCookie)
    			Response.write("<tr>")
    			Response.write("<th width=""30%"" bgcolor=""#99CDFF"" align=""left"">" & strCookie & "(" & strKey & ")</th>")
    			Response.write("<td bgcolor=""#CBF3FF"" align=""left"">" & Request.Cookies(strCookie)(strKey) & " </td>")
    			Response.write("</tr>")
    		Next
    	End if
    Next
    %>
			</table>
			<br />
			<hr />
			<br />
			<h3>Session Cookies</h3>
			<table width="100%" border="0" bgcolor="#000000" cellspacing="1" cellpadding="3">
    <%
    For Each strSession In Session.Contents
    		Response.write("<tr>")
    		Response.write("<th width=""30%"" bgcolor=""#99CDFF"" align=""left"">" & strSession & "</th>")
    		Response.write("<td bgcolor=""#CBF3FF"" align=""left"">" & Session(strSession) & " </td>")
    		Response.write("</tr>")
    Next
    %>
			</table>
			<br />
			<hr />
			<br />
			<h3>Other variables</h3>
			<table width="100%" border="0" bgcolor="#000000" cellspacing="1" cellpadding="3">
			<tr><th width="30%" bgcolor="#99CDFF" align="left">Session.sessionid</th><td bgcolor="#CBF3FF"><%= Session.sessionid %></td></tr>
			<tr><th width="30%" bgcolor="#99CDFF" align="left">Server.MapPath</th><td bgcolor="#CBF3FF"><%= Server.MapPath ("/") %></td></tr>
			</table>
			</div>
    <%
End Sub

%>
